VERSION 5.00
Object = "{0ECD9B60-23AA-11D0-B351-00A0C9055D8E}#6.0#0"; "mshflxgd.ocx"
Object = "{0D452EE1-E08F-101A-852E-02608C4D0BB4}#2.0#0"; "FM20.DLL"
Begin VB.UserControl UclCliente 
   BackStyle       =   0  'Transparent
   ClientHeight    =   4200
   ClientLeft      =   0
   ClientTop       =   0
   ClientWidth     =   8415
   ScaleHeight     =   4200
   ScaleWidth      =   8415
   Begin VB.TextBox TxtNit 
      Height          =   285
      Left            =   1080
      MaxLength       =   15
      TabIndex        =   0
      Top             =   0
      Width           =   1455
   End
   Begin VB.TextBox TxtNombre 
      Height          =   285
      Left            =   1080
      MaxLength       =   100
      TabIndex        =   2
      Top             =   360
      Width           =   5655
   End
   Begin VB.CommandButton CmdBuscar 
      Height          =   285
      Left            =   6840
      Picture         =   "UclCliente.ctx":0000
      Style           =   1  'Graphical
      TabIndex        =   3
      ToolTipText     =   "Buscar"
      Top             =   360
      Width           =   375
   End
   Begin VB.TextBox TxtAsoNro 
      Height          =   285
      Left            =   3840
      MaxLength       =   15
      TabIndex        =   1
      Top             =   0
      Visible         =   0   'False
      Width           =   1455
   End
   Begin VB.Frame FraBusqueda 
      BackColor       =   &H00FFFFC0&
      BorderStyle     =   0  'None
      Height          =   3495
      Left            =   0
      TabIndex        =   8
      Top             =   720
      Width           =   8415
      Begin MSHierarchicalFlexGridLib.MSHFlexGrid HFG 
         Height          =   2775
         Left            =   120
         TabIndex        =   4
         Top             =   120
         Width           =   8175
         _ExtentX        =   14420
         _ExtentY        =   4895
         _Version        =   393216
         Cols            =   3
         FixedCols       =   0
         SelectionMode   =   1
         AllowUserResizing=   1
         _NumberOfBands  =   1
         _Band(0).Cols   =   3
      End
      Begin MSForms.CommandButton CmdAdicionar 
         Height          =   375
         Left            =   120
         TabIndex        =   7
         Top             =   3000
         Width           =   1455
         Caption         =   " Adicionar"
         PicturePosition =   327683
         Size            =   "2566;661"
         Picture         =   "UclCliente.ctx":058A
         Accelerator     =   65
         FontHeight      =   165
         FontCharSet     =   0
         FontPitchAndFamily=   2
         ParagraphAlign  =   3
      End
      Begin MSForms.CommandButton CmdCancelar 
         Height          =   375
         Left            =   5160
         TabIndex        =   6
         Top             =   3000
         Width           =   1455
         Caption         =   " Cancelar"
         PicturePosition =   327683
         Size            =   "2566;661"
         Picture         =   "UclCliente.ctx":06E4
         Accelerator     =   67
         FontHeight      =   165
         FontCharSet     =   0
         FontPitchAndFamily=   2
         ParagraphAlign  =   3
      End
      Begin MSForms.CommandButton CmdSeleccionar 
         Height          =   375
         Left            =   6840
         TabIndex        =   5
         Top             =   3000
         Width           =   1455
         Caption         =   " Seleccionar"
         PicturePosition =   327683
         Size            =   "2566;661"
         Picture         =   "UclCliente.ctx":083E
         Accelerator     =   83
         FontHeight      =   165
         FontCharSet     =   0
         FontPitchAndFamily=   2
         ParagraphAlign  =   3
      End
      Begin VB.Line LinIzq 
         BorderColor     =   &H00FF6666&
         BorderWidth     =   6
         X1              =   0
         X2              =   0
         Y1              =   0
         Y2              =   3480
      End
      Begin VB.Line LinDer 
         BorderColor     =   &H00FF6666&
         BorderWidth     =   5
         X1              =   8400
         X2              =   8400
         Y1              =   0
         Y2              =   3480
      End
      Begin VB.Line LinArr 
         BorderColor     =   &H00FF6666&
         BorderWidth     =   6
         X1              =   0
         X2              =   8400
         Y1              =   0
         Y2              =   0
      End
      Begin VB.Line LinAba 
         BorderColor     =   &H00FF6666&
         BorderWidth     =   5
         X1              =   0
         X2              =   8400
         Y1              =   3465
         Y2              =   3465
      End
   End
   Begin VB.Label LabNit 
      Alignment       =   1  'Right Justify
      Caption         =   "Nit o C.I.:"
      Height          =   255
      Left            =   0
      TabIndex        =   11
      Top             =   0
      Width           =   975
   End
   Begin VB.Label LabNombre 
      Alignment       =   1  'Right Justify
      Caption         =   "Cliente:"
      Height          =   195
      Left            =   0
      TabIndex        =   10
      Top             =   360
      Width           =   1005
   End
   Begin VB.Label LabAsoNro 
      Alignment       =   1  'Right Justify
      Caption         =   "Nro. Asociado:"
      Height          =   255
      Left            =   2640
      TabIndex        =   9
      Top             =   0
      Visible         =   0   'False
      Width           =   1095
   End
End
Attribute VB_Name = "UclCliente"
Attribute VB_GlobalNameSpace = False
Attribute VB_Creatable = True
Attribute VB_PredeclaredId = False
Attribute VB_Exposed = False
Option Explicit

Public WithEvents cliente As ClsCliente
Attribute cliente.VB_VarHelpID = -1
Public WithEvents asociado As ClsAsociado
Attribute asociado.VB_VarHelpID = -1
Public WithEvents comisionista As ClsComisionista
Attribute comisionista.VB_VarHelpID = -1
Public WithEvents inscrito As ClsInscrito
Attribute inscrito.VB_VarHelpID = -1

Public Event seleccionandoCliente()

Private estaMaximizado As Boolean
Private opcNuevo As Boolean
Private opcTodos As Boolean
'para la rueda
Private cMouse As New ClsRueda

Private controlActual As Integer
'indica cual es el control sobre el cual se hace enter
'1: txtnroaso
'2: txtnit
'3: txtnombre
Private estaEnviandoTab As Boolean
'para que no envie doble tab

Private alto_small As Integer
Private alto_large As Integer

Private tipoDocumentoCodigo As String

'constructores y destructores
Private Sub UserControl_Initialize()
    cMouse.InitMouseWheel HFG.hwnd
    estaMaximizado = False
    estaEnviandoTab = False
    controlActual = 0
    alto_small = 660
    alto_large = 4200
End Sub

Private Sub UserControl_Terminate()
    Set cliente = Nothing
    Set asociado = Nothing
    Set inscrito = Nothing
    Set comisionista = Nothing
End Sub

'redimensionamiento
Private Sub Achica()
    Height = alto_small
    estaMaximizado = False
End Sub
Private Sub Agranda()
    Height = alto_large
    estaMaximizado = True
End Sub

'inicio de objetos
Public Sub CargarClientes(Optional opcionTodos As Boolean = False, Optional opcionNuevo As Boolean = True)
    Set asociado = Nothing
    Set comisionista = Nothing
    Set inscrito = Nothing
    Set cliente = New ClsCliente
    opcTodos = opcionTodos
    opcNuevo = opcionNuevo
    CmdAdicionar.Visible = opcNuevo
    controlActual = 0
End Sub
Public Sub CargarAsociados(Optional opcionTodos As Boolean = False, Optional opcionNuevo As Boolean = True)
    Set cliente = Nothing
    Set comisionista = Nothing
    Set inscrito = Nothing
    Set asociado = New ClsAsociado
    opcTodos = opcionTodos
    opcNuevo = opcionNuevo
    CmdAdicionar.Visible = opcNuevo
    controlActual = 0
    'muestra y reubica el label y textbox de nroregistro asociados
End Sub
Public Sub CargarComisionistas(tipoDocCod As String, Optional opcionTodos As Boolean = False, Optional opcionNuevo As Boolean = True)
    Set asociado = Nothing
    Set cliente = Nothing
    Set inscrito = Nothing
    Set comisionista = New ClsComisionista
    
    tipoDocumentoCodigo = tipoDocCod
    opcTodos = opcionTodos
    opcNuevo = opcionNuevo
    CmdAdicionar.Visible = opcNuevo
    controlActual = 0
    'muestra y reubica el label y textbox de nroregistro asociados
End Sub
Public Sub CargarInscritos(Optional opcionTodos As Boolean = False, Optional opcionNuevo As Boolean = True)
    Set asociado = Nothing
    Set cliente = Nothing
    Set comisionista = Nothing
    Set inscrito = New ClsInscrito
    
    opcTodos = opcionTodos
    opcNuevo = opcionNuevo
    CmdAdicionar.Visible = opcNuevo
    controlActual = 0
    'muestra y reubica el label y textbox de nroregistro asociados
End Sub
'BUSQUEDA GENERICA
Private Sub buscar()
    If Not cliente Is Nothing Then
        BuscarCliente
    ElseIf Not asociado Is Nothing Then
        BuscarAsociado
    ElseIf Not comisionista Is Nothing Then
        BuscarComisionista
    ElseIf Not inscrito Is Nothing Then
        BuscarInscrito
    End If
End Sub
'busqueda de cliente
Private Sub BuscarCliente()
    cliente.buscarGrid TxtNit.Text, txtnombre.Text
    If cliente.orset.numRegistros = 1 Then
        'solo encuentra UN CLIENTE => lo selecciona
        '1.- lo coloca en la grilla
        Set HFG.Recordset = cliente.orset.mrec
        'lo configura
        configuraGrid
        mostrarDatos
        RaiseEvent seleccionandoCliente
        'se sale del control
        SalirControl
    ElseIf cliente.orset.numRegistros > 1 Then
        'encuentra MAS DE UN CLIENTE => los muestra en la lista
        Set HFG.Recordset = cliente.orset.mrec
        configuraGrid
        Agranda
        'se ubica en el grid
        HFG.SetFocus
    Else
        'no encuentra NINGUN CLIENTE
        HFG.Clear
        HFG.Cols = cliente.orset.numCampos
        HFG.Rows = 2
        configuraGrid
        Agranda
        If controlActual = 1 Then
            TxtNit.SetFocus
        ElseIf controlActual = 2 Then
            txtnombre.SetFocus
        ElseIf controlActual = 3 Then
            CmdAdicionar.SetFocus
        End If
        controlActual = 0
    End If
End Sub
'busqueda de asociados
Private Sub BuscarAsociado()
    asociado.buscarGrid TxtNit.Text, txtnombre.Text, TxtAsoNro.Text
    If asociado.orset.numRegistros = 1 Then
        'solo encuentra UN CLIENTE => lo selecciona
        '1.- lo coloca en la grilla
        Set HFG.Recordset = asociado.orset.mrec
        'lo configura
        configuraGrid
        mostrarDatos
        RaiseEvent seleccionandoCliente
        'y se sale del control
        SalirControl
    ElseIf asociado.orset.numRegistros > 1 Then
        'encuentra MAS DE UN CLIENTE => los muestra en la lista
        Set HFG.Recordset = asociado.orset.mrec
        configuraGrid
        Agranda
        'se ubica en el grid
        HFG.SetFocus
    Else
        'no encuentra NINGUN ASOCIADO
        HFG.Clear
        HFG.Cols = asociado.orset.numCampos
        HFG.Rows = 2
        configuraGrid
        Agranda
        If controlActual = 1 Then
            TxtNit.SetFocus
        ElseIf controlActual = 2 Then
            txtnombre.SetFocus
        ElseIf controlActual = 3 Then
            CmdAdicionar.SetFocus
        End If
        controlActual = 0
    End If
End Sub

Private Sub BuscarComisionista()
    comisionista.buscarGrid tipoDocumentoCodigo, TxtNit.Text, txtnombre.Text
    If comisionista.orset.numRegistros = 1 Then
        'solo encuentra UN CLIENTE => lo selecciona
        '1.- lo coloca en la grilla
        Set HFG.Recordset = comisionista.orset.mrec
        'lo configura
        configuraGrid
        mostrarDatos
        RaiseEvent seleccionandoCliente
        'se sale del control
        SalirControl
    ElseIf comisionista.orset.numRegistros > 1 Then
        'encuentra MAS DE UN CLIENTE => los muestra en la lista
        Set HFG.Recordset = comisionista.orset.mrec
        configuraGrid
        Agranda
        'se ubica en el grid
        HFG.SetFocus
    Else
        'no encuentra NINGUN CLIENTE
        HFG.Clear
        HFG.Cols = comisionista.orset.numCampos
        HFG.Rows = 2
        configuraGrid
        Agranda
        If controlActual = 1 Then
            TxtNit.SetFocus
        ElseIf controlActual = 2 Then
            txtnombre.SetFocus
        ElseIf controlActual = 3 Then
            CmdAdicionar.SetFocus
        End If
        controlActual = 0
    End If
End Sub
Private Sub BuscarInscrito()
    inscrito.buscarGrid TxtNit.Text, txtnombre.Text
    If inscrito.orset.numRegistros = 1 Then
        'solo encuentra UN CLIENTE => lo selecciona
        '1.- lo coloca en la grilla
        Set HFG.Recordset = inscrito.orset.mrec
        'lo configura
        configuraGrid
        mostrarDatos
        RaiseEvent seleccionandoCliente
        'se sale del control
        SalirControl
    ElseIf inscrito.orset.numRegistros > 1 Then
        'encuentra MAS DE UN CLIENTE => los muestra en la lista
        Set HFG.Recordset = inscrito.orset.mrec
        configuraGrid
        Agranda
        'se ubica en el grid
        HFG.SetFocus
    Else
        'no encuentra NINGUN CLIENTE
        HFG.Clear
        HFG.Cols = inscrito.orset.numCampos
        HFG.Rows = 2
        configuraGrid
        Agranda
        If controlActual = 1 Then
            TxtNit.SetFocus
        ElseIf controlActual = 2 Then
            txtnombre.SetFocus
        ElseIf controlActual = 3 Then
            CmdAdicionar.SetFocus
        End If
        controlActual = 0
    End If
End Sub

'metodos de la grilla
Private Sub configuraGrid()
    Dim i As Integer
    'siempre debe configurarse despues de hacer la busqueda
    If Not cliente Is Nothing Then
    'clientes
        '0: cli_id, 0px
        '1: cli_nit, 100px
        '2: cli_nombre, 500px
        For i = 0 To cliente.orset.numCampos - 1
            HFG.TextMatrix(0, i) = cliente.orset.getCampo(i)
            Select Case HFG.TextMatrix(0, i)
                Case cli_nit: HFG.ColWidth(i) = 100 * 15
                              HFG.TextMatrix(0, i) = "Nit"
                Case cli_nombre: HFG.ColWidth(i) = 400 * 15
                                 HFG.TextMatrix(0, i) = "Nombre"
                Case Else: HFG.ColWidth(i) = 0 * 15
            End Select
        Next
    ElseIf Not asociado Is Nothing Then
    'asociados
        '0: cli_id, 0px
        '1: aso_nro_registro, 100px
        '2: aso_nit, 100px
        '3: aso_nombre, 300px
        For i = 0 To asociado.orset.numCampos - 1
            HFG.TextMatrix(0, i) = asociado.orset.getCampo(i)
            Select Case HFG.TextMatrix(0, i)
                Case aso_nro_registro: HFG.ColWidth(i) = 100 * 15
                                       HFG.TextMatrix(0, i) = "Nro.Reg."
                Case aso_nit: HFG.ColWidth(i) = 100 * 15
                              HFG.TextMatrix(0, i) = "Nit"
                Case aso_nombre: HFG.ColWidth(i) = 300 * 15
                                 HFG.TextMatrix(0, i) = "Nombre"
                Case Else: HFG.ColWidth(i) = 0 * 15
            End Select
        Next
    ElseIf Not comisionista Is Nothing Then
    'clientes
        '0: cli_id, 0px
        '1: cli_nit, 100px
        '2: cli_nombre, 500px
        For i = 0 To comisionista.orset.numCampos - 1
            HFG.TextMatrix(0, i) = comisionista.orset.getCampo(i)
            Select Case HFG.TextMatrix(0, i)
                Case com_nit: HFG.ColWidth(i) = 100 * 15
                              HFG.TextMatrix(0, i) = "Nit"
                Case com_nombre: HFG.ColWidth(i) = 400 * 15
                                 HFG.TextMatrix(0, i) = "Nombre"
                Case Else: HFG.ColWidth(i) = 0 * 15
            End Select
        Next
    ElseIf Not inscrito Is Nothing Then
    'clientes
        '0: cli_id, 0px
        '1: cli_nit, 100px
        '2: cli_nombre, 500px
        For i = 0 To inscrito.orset.numCampos - 1
            HFG.TextMatrix(0, i) = inscrito.orset.getCampo(i)
            Select Case HFG.TextMatrix(0, i)
                Case ins_ci: HFG.ColWidth(i) = 100 * 15
                              HFG.TextMatrix(0, i) = "C.I."
                Case ins_nombre: HFG.ColWidth(i) = 400 * 15
                                 HFG.TextMatrix(0, i) = "Nombre"
                Case Else: HFG.ColWidth(i) = 0 * 15
            End Select
        Next
    End If
End Sub

Private Sub mostrarDatos()
    If Not cliente Is Nothing Then
        TxtNit.Text = cliente.nit
        txtnombre.Text = cliente.nombre
    ElseIf Not asociado Is Nothing Then
        TxtNit.Text = asociado.nit
        txtnombre.Text = asociado.nombre
        TxtAsoNro.Text = asociado.nro_registro
    ElseIf Not comisionista Is Nothing Then
        TxtNit.Text = comisionista.nit
        txtnombre.Text = comisionista.nombre
    ElseIf Not inscrito Is Nothing Then
        TxtNit.Text = inscrito.ci
        txtnombre.Text = inscrito.nombre
    End If
End Sub


'SETTER DE CLIENTE Y ASOCIADO
Public Sub setCliente(oCliente As ClsCliente, Optional dummyraise As Boolean = False)
    Set cliente = oCliente
    TxtNit.Text = cliente.nit
    txtnombre.Text = cliente.nombre
End Sub
Public Sub setComisionista(oComisionista As ClsCliente)
    Set comisionista = oComisionista
    TxtNit.Text = comisionista.nit
    txtnombre.Text = comisionista.nombre
End Sub
Public Sub setInscrito(oInscrito As ClsInscrito)
    Set inscrito = oInscrito
    TxtNit.Text = inscrito.ci
    txtnombre.Text = inscrito.nombre
End Sub

Public Sub deseleccionarCliente()
    If Not cliente Is Nothing Then
        CargarClientes opcTodos, opcNuevo
    ElseIf Not asociado Is Nothing Then
        CargarAsociados opcTodos, opcNuevo
    ElseIf Not asociado Is Nothing Then
        CargarComisionistas tipoDocumentoCodigo, opcTodos, opcNuevo
    ElseIf Not inscrito Is Nothing Then
        CargarInscritos opcTodos, opcNuevo
    End If
    TxtAsoNro.Text = ""
    TxtNit.Text = ""
    txtnombre.Text = ""
End Sub
'/////////////////////////////////
'   emulando propiedades grales
'/////////////////////////////////
Public Sub enabled(isEnabled As Boolean)
    TxtNit.enabled = isEnabled
    txtnombre.enabled = isEnabled
    TxtAsoNro.enabled = isEnabled
    CmdBuscar.enabled = isEnabled
End Sub

Private Sub SalirControl()
    Achica
    CmdAdicionar.SetFocus
    'Call CmdAdicionar_GotFocus
    If Not estaEnviandoTab Then
        SendKeys "{Tab}", 0
    End If
    estaEnviandoTab = False
End Sub

Public Sub setLabels(nitLabel As String, nombreLabel As String)
    LabNit.Caption = nitLabel
    LabNombre.Caption = nombreLabel
End Sub

Public Sub setTamano(Optional ancho As Integer = 0, Optional alto As Integer = 0)
    If ancho > 0 Then
        FraBusqueda.Width = ancho
        LinDer.x1 = ancho - 15
        LinDer.x2 = ancho - 15
        CmdSeleccionar.Left = 6840 - (8415 - ancho)
        CmdCancelar.Left = 5160 - (8415 - ancho)
        HFG.Width = 8175 - (8415 - ancho)
        CmdBuscar.Left = 6840 - (8415 - ancho)
        txtnombre.Width = 5655 - (8415 - ancho)
    End If
    If alto > 0 Then
        alto_large = alto
        FraBusqueda.Height = 3495 - (4200 - alto)
        LinAba.Y1 = 3465 - (4200 - alto)
        LinAba.Y2 = 3465 - (4200 - alto)
        HFG.Height = 2775 - (4200 - alto)
        CmdSeleccionar.Top = 3000 - (4200 - alto)
        CmdCancelar.Top = 3000 - (4200 - alto)
        CmdAdicionar.Top = 3000 - (4200 - alto)
    End If
End Sub

'//////////////////////////////
'   eventos de los controles
'//////////////////////////////
'seleccionando todo el texto al recibir el foco
Private Sub TxtAsoNro_GotFocus()
    seleccionaTexto TxtAsoNro
End Sub
Private Sub TxtNit_GotFocus()
    seleccionaTexto TxtNit
End Sub
Private Sub TxtNombre_GotFocus()
    seleccionaTexto txtnombre
End Sub

Private Sub TxtAsoNro_KeyPress(KeyAscii As Integer)
    Dim Kt As Integer
    Kt = KeyAscii
    KeyAscii = NroEntero(KeyAscii)
    If Kt = vbKeyReturn Then
        estaEnviandoTab = True
        controlActual = 1
        buscar
    End If
End Sub
Private Sub TxtNit_KeyPress(KeyAscii As Integer)
    Dim Kt As Integer
    Kt = KeyAscii
    KeyAscii = NroEntero(KeyAscii)
    If Kt = vbKeyReturn Then
        estaEnviandoTab = True
        controlActual = 2
        buscar
    End If
End Sub
Private Sub txtnombre_KeyPress(KeyAscii As Integer)
    Dim Kt As Integer
    Kt = KeyAscii
    KeyAscii = mayusculas(KeyAscii)
    If Kt = vbKeyReturn Then
        estaEnviandoTab = True
        controlActual = 3
        buscar
    End If
End Sub

'buscando al cliente al hacer click en buscar
'boton buscar
Private Sub CmdBuscar_Click()
    buscar
End Sub
'Hflexgrid
Private Sub HFG_DblClick()
    Call CmdSeleccionar_Click
End Sub
Private Sub HFG_KeyPress(KeyAscii As Integer)
    If KeyAscii = 13 Then
        Call CmdSeleccionar_Click
    End If
End Sub
'boton seleccionar
Private Sub CmdSeleccionar_Click()
    If Not cliente Is Nothing Then
        cliente.orset.setFila HFG.Row - 1
    ElseIf Not asociado Is Nothing Then
        asociado.orset.setFila HFG.Row - 1
    ElseIf Not comisionista Is Nothing Then
        comisionista.orset.setFila HFG.Row - 1
    ElseIf Not inscrito Is Nothing Then
        inscrito.orset.setFila HFG.Row - 1
    End If
    mostrarDatos
    Achica
    RaiseEvent seleccionandoCliente
    CmdAdicionar.SetFocus
End Sub
'boton cancelar
Private Sub CmdCancelar_Click()
    Achica
End Sub
'boton adicionar
Private Sub CmdAdicionar_Click()
    'si se puede adicionar registros lo adiciona
    If opcNuevo Then
        'lo inserta en la bd
        If Not cliente Is Nothing Then
            cliente.Nuevo TxtNit.Text, txtnombre.Text, "1", "0"
        ElseIf Not asociado Is Nothing Then
            'crea un nuevo asociado
            'asociado.Nuevo TxtNit.Text, TxtNombre.Text, "1", "0"
        ElseIf Not comisionista Is Nothing Then
            comisionista.Nuevo tipoDocumentoCodigo, TxtNit.Text, txtnombre.Text
        ElseIf Not inscrito Is Nothing Then
            inscrito.Nuevo TxtNit.Text, txtnombre.Text
        End If
        'ya esta seleccionado por objetos
        'muestra sus datos en pantalla
        mostrarDatos
        'se sale del control
        estaEnviandoTab = False
        SalirControl
    End If
End Sub

'para cuando reciben el foco los controles que pueden estar ocultos
Private Sub HFG_GotFocus()
    If Not estaMaximizado Then
        SendKeys "{Tab}", 0
    End If
End Sub
Private Sub CmdSeleccionar_GotFocus()
    If Not estaMaximizado Then
        SendKeys "{Tab}", 0
    End If
End Sub
Private Sub CmdCancelar_GotFocus()
    If Not estaMaximizado Then
        SendKeys "{Tab}", 0
    End If
End Sub
Private Sub CmdAdicionar_GotFocus()
    If Not estaMaximizado Then
        SendKeys "{Tab}", 0
    End If
End Sub
